

## Tema 1 – Introducción

Fecha de entrega 29/09/21 23:55h

### Actividades para entregar<sup>1</sup>:

1. Desarrollar en pseudocódigo un algoritmo para obtener la cuantía a pagar del recibo del IBI tras retrasarnos en el pago y aplicar un porcentaje de recargo (expresado como tanto por cien). Tanto la cuantía inicial como el porcentaje se introducen por teclado.
2. Desarrollar en pseudocódigo un algoritmo para obtener la calificación global en prácticas de un estudiante sabiendo que se calcula a partir de la suma de las notas de las dos mejores (de tres).
3. Desarrollar en pseudocódigo un algoritmo para sumar una serie de números introducidos por teclado. Si el usuario introduce un número negativo se para el proceso y se escribe el resultado acumulado.
4. Dado un número entero positivo x cualquiera, se aplica la siguiente operación: si x es par se divide entre 2; si x es impar se multiplica por 3 y se suma 1<sup>2</sup>. Al número obtenido aplicar la misma operación. Por ejemplo, a partir del 12, se obtienen 6, 3, 10, 5, 16, 8, 4, 2, 1. Desarrollar en pseudocódigo un programa que escriba dicha secuencia a partir de un número introducido por teclado.
5. El problema de Basilea consiste en encontrar la suma exacta de los inversos de los cuadrados de los enteros positivos:

$$\sum_{i=1}^n 1/i^2 = 1 + \frac{1}{2^2} + \frac{1}{3^2} + \frac{1}{4^2} + \dots + \frac{1}{n^2}$$

A partir del algoritmo sustituir la estructura de repetición *Para...hasta...hacer*, por *repetir...hasta que* y por *mientras...hacer*.

```
Algoritmo Basilea
  Escribir "Intro límite n > 0"
  Leer n
  suma <- 0
  Para i<- 1 Hasta n Con Paso 1 Hacer
    suma <- suma + 1/i^2
  Fin Para
  Escribir suma
FinAlgoritmo
```

---

<sup>1</sup> Puede usarse la sintaxis de PSeInt

<sup>2</sup> La conjetura de Collatz dice que, sea cual sea el número x inicial, tras un número finito de repeticiones de la operación, se llega a 1.

## Soluciones

1.

```
Algoritmo recargo
  leer recibo
  leer porcentaje
  Escribir recibo+recibo * porcentaje/100
FinAlgoritmo
```

2.

```
Algoritmo calificaciones
  Escribir "Intro A, B y C"
  Leer A, B, C
  Si A>=C y B>=C Entonces
    total <- A + B
  SiNo
    Si B>=A y C>=A Entonces
      total <- C + B
    SiNo
      total <- A + C
    Fin Si
  Fin Si
  Escribir "Nota es ", total
FinAlgoritmo
```

3.

```
Algoritmo suma_serie
  numero<-0
  suma<-0
  Repetir
    leer numero
    Si numero >= 0 Entonces
      suma<-suma+numero
    Fin Si
  Hasta Que numero<=0
  escribir suma
FinAlgoritmo
```

4.

```
Algoritmo Collatz
  Escribir "Intro número "
  Leer n
  Repetir
    Escribir n
    Si n mod 2 = 0 Entonces
      n <- n / 2
    SiNo
      n <- n * 3 + 1
    FinSi
  Hasta Que n = 1
  Escribir n
FinAlgoritmo
```

5.

```
suma <- 0
i <- 1
Repetir
  suma <- suma + 1/i^2
  i <- i + 1
Hasta Que i>n
Escribir suma
```

```
suma<-0
i<-1
Mientras i<=n Hacer
  suma <- suma + 1/i^2
  i <- i + 1
Fin Mientras
Escribir suma
```